
pkgname=libmpc
pkgver=1.1.0
pkgrel=1
pkgdesc="Library for the arithmetic of complex numbers with arbitrarily high precision"
arch=('x86_64')
url="http://www.multiprecision.org"
license=('LGPL')
depends=('mpfr')
options=('!libtool')
source=("https://ftp.gnu.org/gnu/mpc/mpc-${pkgver}.tar.gz")
md5sums=('4125404e41e482ec68282a2e687f6c73')

build() {
  cd mpc-${pkgver}
  
  ./configure --prefix=/usr
  make
}

check() {
  cd mpc-${pkgver}
  
  make check
}

package() {
  cd mpc-${pkgver}
  make DESTDIR=${pkgdir} install
  
  mv ${pkgdir}/usr/share/info/{mpc,libmpc}.info
}
